

富邦 API 整合式下單系統說明

□系統執行流程與程序

本範例的報價取得部分，採用富邦期貨的 API 或 RTD 或 DDE，下單部分採用富邦期貨下單 API。系統使用前必須先於 Excel VBA 中設定引用項目(「工具」➤「設定引用項目」)，如圖 1 所示。




圖 1 於 Excel 中設定引用項目

並打開富邦 E01 系統。



圖 2 打開富邦 E01

若欲執行下單則須執行 TransferIP 系統作 IP 轉換。(這部分功能係由寶碩資訊提供)




圖 3 打開富邦 E01

接下來可打開「FubonE01_AutoTrading_DDEQuote&APIOrder_V1」或「FubonE01_AutoTrading_APIQuote&APIOrder_V1」檔，分別可接收 DDE 與 RTD 報價資料與 API 報價資料。進入系統後若使用 DDE 或 API 取得報價必須啟用自動更新連結，如圖 4。




圖 4 啟用自動更新連結

如此在「Quote_Data」工作表中即可連上即時資料，如圖 5 所示。




圖 5(a) DDE 與 RTD 報價




圖 5(b) API 報價

進入系統後可先進入圖 6「Order 工作表」建立連線。

最後進入「K_Bar 工作表」按鍵啟動「啟動 DDE 自動交易系統」按鍵，即可開始進行自動下單交易(下模擬單)，即時下單過程如圖 6 所示。




圖 5 Order 工作表




圖 6 即時下單過程

下單的結果可以在「富邦 E01」中看到模擬下單結果(如圖 7)，實際下單僅需重新導向作實際下單即可。

期貨委託單											
動作	委託時間	委託單號	買賣別	商品名稱	原委託	價位	倉別	委託條件	取消	成交	成交均價
賣出	12:39:11	8B771	賣出	台股指數11	1	7709.000	新倉	IOC	0	1	8286.000
賣出	11:54:01	8B6F2	賣出	台股指數11	1	7709.000	新倉	IOC	0	1	8285.000
賣出	11:53:01	8B6EA	賣出	台股指數11	1	7709.000	新倉	IOC	0	1	8283.000
賣出	11:52:01	8B6E6	賣出	台股指數11	1	7709.000	新倉	IOC	0	1	8277.000
賣出	11:19:01	8B6C6	賣出	台股指數11	1	7709.000	新倉	IOC	0	1	8261.000
賣出	11:07:16	8B6B4	賣出	台股指數11	1	8869.000	新倉	IOC	0	1	8254.000
賣出	09:21:01	8B570	賣出	台股指數11	1	8869.000	自動	IOC	0	1	8269.000
賣出	09:19:43	8B56F	賣出	台股指數11	1	8869.000	自動	IOC	0	1	8262.000

圖 7 模擬下單結果

系統說明

系統程式碼如下表所示。

行號	程式碼
1	Private Declare Function SetTimer Lib "user32" (ByVal hWnd As Long, ByVal nIDEvent As Long, ByVal uElapsed As Long, ByVal lpTimerFunc As Long) As Long
2	Private Declare Function KillTimer Lib "user32" (ByVal hWnd As Long, ByVal nIDEvent As Long) As Long
3	Public Fubon_Mananger_obj As Object
4	Dim hTimer
5	Dim BarNo As Integer, SheetsRow As Integer, Initial_Flag As Integer

```

6 Dim FMA_P As Integer, SMA_P As Integer, RSI_P As Integer, RSI_B As Single, RSI_S As Single
7 Dim Min_Bar(300, 17) As Single
8 'No, Time, O, H, L, C, V, CV, FMA, SMA, Deci1, CloseChange, RSI, Deci2, FinalD, Return, Accumu. Return
9 Dim O As Single, H As Single, L As Single, C As Single, V As Single, FMA As Single, SMA As Single
10 Dim PreTickTime As Single
11 Dim Position As Integer

12 '關閉 DDE 自動交易系統程序
13 Sub Close_DDE_Auto_Order()
14     On Error Resume Next
15     Call KillTimer(0, hTimer)
16     Exit Sub
17 ErrorHandler:
18     MsgBox Err.Description
19 End Sub

20 '啟動 DDE 自動交易系統程序
21 Sub Start_DDE_Auto_Order()
22     Initial_Flag = 0
23     Position = 0
24     On Error Resume Next
25     hTimer = SetTimer(0, 0, 1 * 1000, AddressOf Main) '單位為毫秒
26     Exit Sub
27 ErrorHandler:
28     MsgBox Err.Description
29 End Sub

30 Private Sub Main()
31     Dim tmp_RGB As Object
32     Dim i As Integer
33     Dim tempTickTime As String, tmpTickTime As String, TickTime As Single
34     Dim TickPrice As Single, TickCVolume As Single
35     On Error Resume Next
36     '=====此例接收 API 報價=====
37     tmpTickTime = Sheets("Quote_Data").Cells(3, 3)
38     tmpTickTime1 = Mid(tmpTickTime, 1, 2) + Mid(tmpTickTime, 4, 2) + Mid(tmpTickTime, 7, 2) '時間格式為 08:45:00
39     TickTime = Int(Val(tmpTickTime1) / 100)
40     TickPrice = Sheets("Quote_Data").Cells(3, 4) '成交價
41     TickCVolume = Sheets("Quote_Data").Cells(3, 5) '累積量
42     '啟始程序

```

```

43 If Initial_Flag = 0 Then
44     BarNo = 1                                '設定 K Bar 編號初值
45     SheetsRow = 1                            '設定列編號初值
46     Position = Sheets("K_Bar").Cells(2, 18)  '取得起始部位
47     '技術指標參數設定區
48     FMA_P = 6                                '設定快速均線參數
49     SMA_P = 10                               '設定慢速均線參數
50     RSI_P = 8                                '設定 RSI 指標回算天數
51     RSI_B = 0.4                              '設定 RSI 指標買進指標
52     RSI_S = 0.6                              '設定 RSI 指標賣出指標
53     PreTickTime = TickTime
54     Min_Bar(BarNo, 1) = BarNo                '設定 K 線編號
55     Min_Bar(BarNo, 2) = TickTime             '設定 K 線時間
56     Min_Bar(BarNo, 3) = TickPrice            '設定以成交價為開盤價
57     Min_Bar(BarNo, 4) = TickPrice            '設定以成交價為最高價
58     Min_Bar(BarNo, 5) = TickPrice            '設定以成交價為最低價
59     Min_Bar(BarNo, 6) = TickPrice            '設定以成交價為收盤價
60     Min_Bar(BarNo, 7) = TickCVolume          '設定以累積量為成交量
61     Min_Bar(BarNo, 8) = TickCVolume          '設定以累積量為累積量
62     Initial_Flag = 1                        '設定完成初始設定
63 End If
64 '此段落產生一分 K 線，其他頻率 K 線可以修改
65 If PreTickTime = TickTime Then              '確定為同一分鐘內之報價
66     If TickPrice > Min_Bar(BarNo, 4) Then Min_Bar(BarNo, 4) = TickPrice '產生新高價
67     If TickPrice < Min_Bar(BarNo, 5) Then Min_Bar(BarNo, 5) = TickPrice '產生新低價
68     Min_Bar(BarNo, 6) = TickPrice            '產生新收價
69     Min_Bar(BarNo, 8) = TickCVolume          '產生新累積量
70     Min_Bar(BarNo, 7) = Min_Bar(BarNo, 8) - Min_Bar(BarNo - 1, 8) '產生新交易量(透過累積量相減取得成交量)
71 Else                                         '確定不為同一分鐘內之報價
72     PreTickTime = TickTime                  '更新 K 線時間
73     SheetsRow = SheetsRow + 1               '更新列編號值
74     '以下段落用以將開高收低量等存於 Min_Bar 陣列中資訊輸出到格位中，若欲增加執行效率可刪除
75     Sheets("K_Bar").Cells(SheetsRow, 1) = Min_Bar(BarNo, 1) '輸出 K 線編號
76     Sheets("K_Bar").Cells(SheetsRow, 2) = Min_Bar(BarNo, 2) '輸出 K 線時間
77     Sheets("K_Bar").Cells(SheetsRow, 3) = Min_Bar(BarNo, 7) '輸出 K 線成交量
78     Sheets("K_Bar").Cells(SheetsRow, 4) = Min_Bar(BarNo, 3) '輸出 K 線開盤價
79     Sheets("K_Bar").Cells(SheetsRow, 5) = Min_Bar(BarNo, 4) '輸出 K 線最高價
80     Sheets("K_Bar").Cells(SheetsRow, 6) = Min_Bar(BarNo, 5) '輸出 K 線最低價
81     Sheets("K_Bar").Cells(SheetsRow, 7) = Min_Bar(BarNo, 6) '輸出 K 線收盤價
82     Sheets("K_Bar").Cells(SheetsRow, 8) = Min_Bar(BarNo, 8) '輸出 K 線累積量

```

```

83 '=====
84 '%%以下段落交易者可使用自己交易策略於盤中觸發交易訊息，在此以均線交叉策略與 RSI 擺盪策略為例%%
85 '長短均線計算與策略觸發=>可以快速計算法加速之
86     Dim Sum As Single
87     If BarNo >= 11 Then
88         Sum = 0
89         For i = BarNo - FMA_P + 1 To BarNo
90             Sum = Sum + Min_Bar(i, 6)
91         Next i
92         Min_Bar(BarNo, 9) = Sum / FMA_P
93         Sum = 0
94         For i = BarNo - SMA_P + 1 To BarNo
95             Sum = Sum + Min_Bar(i, 6)
96         Next i
97         Min_Bar(BarNo, 10) = Sum / SMA_P
98         If BarNo >= 11 Then '保留前 12 筆不交易，亦可讀取作日 1 分鐘 K 線存於陣列中，即可於第一筆交易
99             If (Min_Bar(BarNo - 1, 9) < Min_Bar(BarNo - 1, 10)) And (Min_Bar(BarNo, 9) >= Min_Bar(BarNo, 10)) Then
100                 '黃金交叉判斷
101                 Min_Bar(BarNo, 11) = 1
102             End If
103             If (Min_Bar(BarNo - 1, 9) > Min_Bar(BarNo - 1, 10)) And (Min_Bar(BarNo, 9) <= Min_Bar(BarNo - 1, 10)) Then
104                 '死交叉判斷
105                 Min_Bar(BarNo, 11) = -1
106             End If
107         End If
108     End If
109 'RSI 指標計算與策略觸發=>可以快速計算法加速之
110     Dim RSI_Up_Average As Single, RSI_Up_Count As Integer, RSI_Down_Average As Single, RSI_Down_Count As Integer
111     If BarNo > 1 Then Min_Bar(BarNo, 12) = Min_Bar(BarNo, 6) - Min_Bar(BarNo - 1, 6) '漲跌價計算
112     If BarNo >= 11 Then
113         RSI_Up_Average = 0: RSI_Up_Count = 0: RSI_Down_Average = 0: RSI_Down_Count = 0
114         For i = BarNo - RSI_P + 1 To BarNo
115             If Min_Bar(i, 12) > 0 Then
116                 RSI_Up_Average = RSI_Up_Average + Min_Bar(i, 12)
117                 RSI_Up_Count = RSI_Up_Count + 1
118             End If
119             If Min_Bar(i, 12) < 0 Then
120                 RSI_Down_Average = RSI_Down_Average + Min_Bar(i, 12)
121                 RSI_Down_Count = RSI_Down_Count + 1
122             End If

```



```

123     Next i
124     If RSI_Up_Count > 0 Then RSI_Up_Average = RSI_Up_Average / RSI_Up_Count Else RSI_Up_Average = 0
125     '計算平均漲幅
126     If RSI_Down_Count > 0 Then RSI_Down_Average = Abs(RSI_Down_Average / RSI_Down_Count) Else RSI_Down_Average = 0
127     '計算平均跌幅
128     If (RSI_Up_Average + RSI_Down_Average) > 0 Then '計算 RSI 指標值
129         Min_Bar(BarNo, 13) = RSI_Up_Average / (RSI_Up_Average + RSI_Down_Average)
130     Else
131         Min_Bar(BarNo, 13) = 0
132     End If
133     If Min_Bar(BarNo, 13) <= RSI_B Then Min_Bar(BarNo, 14) = 1        'RSI 指標作多
134     If Min_Bar(BarNo, 13) >= RSI_S Then Min_Bar(BarNo, 14) = -1     'RSI 指標作空
135 End If
136 '綜合交易策略判斷，MA 指標與 RSI 指標綜合決策
137 Min_Bar(BarNo, 16) = Position * (Min_Bar(BarNo, 6) - Min_Bar(BarNo, 3)) '隔 Bar 下單前算當 Bar 損益(假設前收=當開)
138 Min_Bar(BarNo, 17) = Min_Bar(BarNo - 1, 16) + Min_Bar(BarNo, 16)    '計算策略累積損益
139 If Position < 1 And (Min_Bar(BarNo, 11) = 1 And Min_Bar(BarNo, 14) = 1) Then '當均線指標作多"且"RSI 指標作多則作多
140     Min_Bar(BarNo, 15) = 1
141     '=====可以市價作多近月台指期 1 口，連接 API 送單=====
142     Sheets("K_Bar").Cells(SheetsRow, 1).Interior.Color = RGB(255, 255, 187) '買訊顏色標示
143     If Position = 0 Then
144         Call eFOrder(Sheets("DDE_Data").Cells(2, 1).Value, "B", "1") '相當於台指期市價買進 1 口 API 送單
145         Position = 1: Sheets("K_Bar").Cells(2, 18) = Position
146     End If
147     If Position = -1 Then
148         Call eFOrder(Sheets("DDE_Data").Cells(2, 1).Value, "B", "2") '相當於台指期市價買進 2 口 API 送單
149         Position = 1: Sheets("K_Bar").Cells(2, 18) = Position
150     End If
151 End If
152 If Position > -1 And (Min_Bar(BarNo, 11) = -1 Or Min_Bar(BarNo, 14) = -1) Then
153     Min_Bar(BarNo, 15) = -1
154     '=====可以市價作空近月台指期 1 口，連接 API 送單=====
155     Sheets("K_Bar").Cells(SheetsRow, 1).Interior.Color = RGB(187, 255, 255) '賣訊顏色標示
156     If Position = 0 Then
157         Call eFOrder(Sheets("DDE_Data").Cells(2, 1).Value, "S", "1") '相當於台指期市價賣出 1 口 API 送單
158         Position = -1: Sheets("K_Bar").Cells(2, 18) = Position
159     End If
160     If Position = 1 Then
161         Call eFOrder(Sheets("DDE_Data").Cells(2, 1).Value, "S", "2") '相當於台指期市價賣出 2 口 API 送單
162         Position = -1: Sheets("K_Bar").Cells(2, 18) = Position

```

```

163         End If
164     End If
165     '=====
166     '以下段落用以將指標計算過程與交易決策過程資訊輸出到格位中，若欲增加執行效率可刪除
167     Sheets("K_Bar").Cells(SheetsRow, 9) = Format(Min_Bar(BarNo, 9), "####0.0")
168     Sheets("K_Bar").Cells(SheetsRow, 10) = Format(Min_Bar(BarNo, 10), "####0.0")
169     Sheets("K_Bar").Cells(SheetsRow, 11) = Min_Bar(BarNo, 11)
170     Sheets("K_Bar").Cells(SheetsRow, 12) = Min_Bar(BarNo, 12)
171     Sheets("K_Bar").Cells(SheetsRow, 13) = Format(Min_Bar(BarNo, 13), "0.000")
172     Sheets("K_Bar").Cells(SheetsRow, 14) = Min_Bar(BarNo, 14)
173     Sheets("K_Bar").Cells(SheetsRow, 15) = Min_Bar(BarNo, 15)
174     Sheets("K_Bar").Cells(SheetsRow, 16) = Min_Bar(BarNo, 16)
175     Sheets("K_Bar").Cells(SheetsRow, 17) = Min_Bar(BarNo, 17)
176     '開始下一 K 線計算
177     BarNo = BarNo + 1                                '設定下一 K 線編號
178     Min_Bar(BarNo, 1) = BarNo                        '設定 K 線標號
179     Min_Bar(BarNo, 2) = TickTime                     '設定 K 線時間
180     Min_Bar(BarNo, 3) = TickPrice                    '以成交價為開盤價
181     Min_Bar(BarNo, 4) = TickPrice                    '以成交價為最高價
182     Min_Bar(BarNo, 5) = TickPrice                    '以成交價為最低價
183     Min_Bar(BarNo, 6) = TickPrice                    '以成交價為收盤價
184     Min_Bar(BarNo, 8) = TickCVolume                  '以累積量為成交量
185     Min_Bar(BarNo, 7) = Min_Bar(BarNo, 8) - Min_Bar(BarNo - 1, 8) '以累積量差異計算新增成交量
186     End If
187 End Sub
188 ....

```

行(1)與行(2)宣告 SetTime 與 KillTime 函數以使用 Windows API 函式的計時器事件。

行(3)到行(11)作模組層次變數宣告，以讓模組內的程序或函數可以使用。BarNo 為 K 線編號，SheetsRow 為工作表列號，Initial_Flag 為起始旗標變數，FMA_P、SMA_P、RSI_P、RSI_B、RSI_S 分別為快速均線、慢速均線、RSI 回算 K 線數、RSI 買進參數、RSI 賣出參數。Min_Bar 為二維陣列，第一維為 300，其為一天交易 300 秒；第二維大小為 17，分別存編號、時間、開、高、低、收、量、累積量、快速均線、慢速均線、均線決策、K 線收盤價差、RSI 指標、RSI 決策、最終決策、當 K 損益、累積損益。PreTickTime 存前一 TickData 的時間。Position 為部位數。

行(12)到行(19)之程序可由 K_Bar 工作表中的「關閉 DDE 自動交易系統」的按鍵驅動，其中行(15)呼叫的 KillTimer 函式用以關閉計時器事件，並以行(16)的 Exit Sub 跳出程序。其他指令為攔截錯誤語法。

行(20)到行(29)之程序可由 K_Bar 工作表中的「啟動 DDE 自動交易系統」的按鍵驅動，其中行(25)呼叫的 SetTimer 函式用以開啟計時器事件，參數 3 之「1*1000」表 1 毫秒之 1000 倍，即每 1 秒驅動第四個參數之「Main」程序。行(16)的 Exit Sub 跳出程序。其他指令為攔截錯誤語法。行(22)可設定啟始旗標(Initial_Flag)為 0，起始部位(Position)為 0。

行(30)到行(187)之程序為 Main 程序內容。

行(37)從 Quote_Data 工作表之 C3 格位取出由 DDE、RTD 或 API 管道取得的即時「時間」資料。

行(38)藉由字串擷取函數(MID)取出「時間」資料中的數值，例如「08:45:00」轉為「084500」。

行(39)將時間資料除以 100 以取得秒數值，例如將「084500」轉為「0845」。

行(40)從 Quote_Data 工作表之 D3 格位取出由 DDE、RTD 或 API 管道取得的即時「成交價」資料。

行(41)從 Quote_Data 工作表之 E3 格位取出由 DDE、RTD 或 API 管道取得的即時「累積量」資料。

行(43)到行(63)間的程式碼用以設定啟始程序。行(44)設定 K Bar 編號初值，行(45)設定列編號初值，行(46)取得起始部位，行(48)設定快速均線參數，行(49)設定慢速均線參數，行(50)設定 RSI 指標回算天數，行(51)設定 RSI 指標買進指標，行(52)設定 RSI 指標賣出指標。行(53)設定前一逐筆資料時間為此一逐筆資料。行(54)設定 K 線編號，行(55)設定 K 線時間，行(56)設定以成交價為開盤價，行(57)設定以成交價為最高價，行(58)設定以成交價為最低價，行(59)設定以成交價為收盤價，行(60)設定以累積量為成交量，行(61)設定以累積量為累積量，行(62)設定完成初始設定，將 Initial_Flag 設為 1 則其後將不會再執行此段落(行(43)到行(63)間)程式碼。行(65)判斷新的 Tick 資料交易時間是否為下一秒資料，若是，則執行行(66)到行(70)間的程式碼，若否，則執行行(72)到行(186)間的程式碼。

行(66)到行(70)間的程式碼分析如後。行(66)判斷最新的成交價若大於目前的最高價，則將最新成交價設定為目前的最高價。行(67)判斷最新的成交價若小於目前的最低價，則將最新成交價設定為目前的最低價。行(68)以最新成交價作為收盤價。行(69)以最新累積量作為累積量。行(69)以最新累積價減去上一 K 線累積量作為成交量。

行(72)到行(186)間的程式碼分析如後。行(72)用以更新 K 線時間。行(72)用以更新列編號值。行(75)到行(82)間的程式碼用以將開高收低量等存於 Min_Bar 陣列中的資訊輸出到試算表格位中，以檢驗邏輯正確性，若欲增加執行效率可刪除此段落。

行(85)到行(164)間的程式碼以「均線交叉信號與 RSI 指標交易信號，形成交易策略」處理 K 線資料產生交易信號，並以富邦證券 API 模擬下單。此段落交易者可以自行決定交易策略並編碼處理 K 線交易資料，同樣也可以連接不同 API 下單函式。

行(87)到行(108)間的程式碼計算長短均線參數並觸發交易信號(黃金交叉作多，死亡交叉作空)；行(88)到行(92)間程式碼計算快速均線參數，行(94)到行(97)間程式碼計算慢速均線參數，行(99)為黃金交叉判斷，行(101)為當黃金交叉訊號出現設定作多(1)，行(103)為死亡交叉判斷，行(105)為當死亡交叉訊號出現設定作空(-1)。

行(109)到行(134)間的程式碼計算 RSI 指標並觸發交易信號(RSI 值過高作空，過低作多)；行(111)作 K 線收盤漲跌價計算，行(114)到行(127)間程式碼計算回算天數期間計算平均漲幅(RSI_Up_Average)與平均跌幅(RSI_Down_Average)。行(128)到行(132)程式碼計算 RSI 指標值。行(133)判斷 RSI 指標過低(\leq RSI_B)則作多，行(134)判斷 RSI 指標過低(\leq RSI_B)則作空。

行(137)計算當一 K 線之損益，即若當一 K 線作多一口，且價格由 8000 點上漲到 8010 點，則賺了 10 點，反之若作空一口，則賠了 10 點(在此我們假設前一 K 線收盤價等於此一 K 線開盤價)。行(138)計算策略累積損益。

行(139)到行(151)間的程式碼依據當「部位(Position) < 1」且「當均線指標作多且 RSI 指標作多」，則作多。「部位(Position) < 1」的條件係為了避免連續作多超過 1 口。行(142)設定買訊產生時標示顏色。

行(143)判斷目前部位(Position)為 0 時，於行(144)以 API 函式(eFOrder)呼叫下單，參數 1 為下單商品(現為台指期)，參數 2 設定買進("B")，參數 3 表下單 1 口("1")。行(145)設定下單後的位置值(Position)並將之顯示於試算表格位中。行(147)判斷目前部位(Position)為 -1 時，於行(148)以 API 函式(eFOrder)呼叫下單，參數 1 為下單商品(現為台指期)，參數 2 設定買進("B")，參數 3 表下單 2 口("2")。行(149)設定下單後的位置值(Position)並將之顯示於試算表格位中。

行(152)到行(164)間的程式碼依據當「部位(Position) > -1」且「當均線指標作空或 RSI 指標作空」，則作空。「部位(Position) > -1」的條件係為了避免連續作空超過 1 口。行(155)設定賣訊產生時標示顏色。

行(156)判斷目前部位(Position)為 0 時，於行(157)以 API 函式(eFOrder)呼叫下單，參數 1 為下單商品(現為台指期)，參數 2 設定賣出("S")，參數 3 表下單 1 口("1")。行(158)設定下單後的位置值(Position)並將之顯示於試算表格位中。行(160)判斷目前部位(Position)為 1 時，於行(161)以 API 函式(eFOrder)呼叫下單，參數 1 為下單商品(現為台指期)，參數 2 設定賣出("S")，參數 3 表下單 2 口("2")。行(162)設定下單後的位置值(Position)並將之顯示於試算表格位中。

行(166)到行(175)間的程式碼用以將指標計算過程與交易決策過程資訊輸出到格位中，若欲增加執行效率可刪除。

行(176)到行(185)間的程式碼用以準備下一 K 線計算。